Corrigé Examen 2 ALGO 12-13  

 SOIT
       A UN ARB ;

       In2 UNE ACTION ;

       Afficher_liste UNE ACTION ;

       Niveau , Distance DES ACTIONS ;

       Niveau_feuille UNE FONCTION ( ENTIER ) ;

       Distance_feuille UNE FONCTION ( ENTIER ) ;

       L UNE LISTE DE ARB ;

       Prec , Q DES POINTEURS VERS UNE LISTE DE ARB ;
      Définir préalablement la structure du bloc de données et du bloc d’en-tête.
       SOIT F UN FICHIER DE ( ENTIER , VECTEUR ( 3 ) ) ENTETE ( ENTIER ) BUFFER Buf ;

      { entete désigne le nombre de blocs}
       Sauvegarde UNE ACTION ;

       Afficher_fichier UNE ACTION ;

   DEBUT
       CREER_ARB ( A , [ 22 , 12 , 56 , 54 , 78 , 67 , 87 ] ) ;

       L := NIL ;

       APPEL In2 ( A ) ;

       APPEL Afficher_liste ( L ) ;

       ECRIRE ( ' les niveaux des feuilles' ) ;

       APPEL Niveau ( L ) ;

       ECRIRE ( ' les distance des feuilles' ) ;

       APPEL Distance ( L ) ;

       OUVRIR ( F , 'test.txt' , 'N' ) ;

       AFF_STRUCT ( Buf , 1 , 0 ) ;

       AFF_ENTETE ( F , 1 , 0 ) ;

       APPEL Sauvegarde ( A ) ;

       ECRIRESEQ ( F , Buf ) ;

       AFF_ENTETE ( F , 1 , ENTETE ( F , 1 ) + 1 ) ;

       FERMER ( F ) ;

       ECRIRE ( ' Afficher le fichier' ) ;

       APPEL Afficher_fichier ;

   FIN
   ACTION Afficher_liste ( A )

   SOIT
       A UNE LISTE DE ARB ;

   DEBUT
       SI A <> NIL
           ECRIRE ( INFO ( VALEUR ( A ) ) ) ;

           APPEL Afficher_liste ( SUIVANT ( A ) ) ;

       FSI
   FIN
Utiliser un parcours récursif pour ranger les  adresses des feuilles d’un arbre de recherche binaire de gauche à droite dans une liste linéaire chaînée. 

ACTION In2 ( A )

   SOIT
       A UN ARB ;

   DEBUT
       SI A <> NIL
           APPEL In2 ( FG ( A ) ) ;
           SI ( FG ( A ) = NIL ) ET ( FD ( A ) = NIL )

               ALLOUER ( Q ) ;

               AFF_VAL ( Q , A ) ;

               AFF_ADR ( Q , NIL ) ;

               SI L = NIL :

                   L := Q

               SINON
                   AFF_ADR ( Prec , Q )

               FSI ;

               Prec := Q ;

           FSI ;

           APPEL In2 ( FD ( A ) )

       FSI
   FIN
Utiliser l’opération Père pour déterminer et afficher le niveau de chaque feuille (élément de la liste créée en 1.)

   ACTION Niveau ( L )

   SOIT
       L , P DES LISTES DE ARB ;

       F UN ARB ;

   DEBUT
       P := L ;

       TQ P <> NIL
           ECRIRE ( Niveau_feuille ( VALEUR ( P ) ) ) ;

           P := SUIVANT ( P )

       FTQ
   FIN
   FONCTION Niveau_feuille ( A ) : ENTIER
   SOIENT
       A , P DES ARB ;

       Niveau UN ENTIER ;

   DEBUT
       Niveau := 0 ;

       P := A ;

       TQ PERE ( A ) <> NIL
           Niveau := Niveau + 1 ;

           A := PERE ( A )

       FTQ ;

       Niveau_feuille := Niveau

   FIN
Utiliser l’opération Père pour déterminer et afficher la distance de chaque feuille (élément de la liste créée en 1.) vers la feuille suivante. La distance désigne le nombre de nœuds traversés.

   FONCTION Distance_feuille ( A ) : ENTIER
   SOIENT
       A , P DES ARB ;

       Distance UN ENTIER ;

       Continue : BOOLEEN ;

   DEBUT
       P := PERE ( A ) ;

       Distance := 1 ;

       Continue := VRAI ;

      {Remonter dans l'arbre}
       TQ Continue

           SI P = NIL
               Continue := FAUX
           SINON
               SI ( FG ( P ) = A ) ET ( FD ( P ) <> NIL )

                   Continue := FAUX ;

               SINON
                   Distance := Distance + 1 ;

                   A := P ;

                   P := PERE ( A )

               FSI
           FSI
       FTQ ;

      { Descendre }
       SI P <> NIL
           P := FD ( P ) ;

           Distance := Distance + 1 ;

           Continue := FAUX ;

           TQ NON Continue

               TQ FG ( P ) <> NIL
                   P := FG ( P ) ;

                   Distance := Distance + 1 ;

               FTQ ;

               SI FD ( P ) = NIL
                   Continue := VRAI
               SINON
                   P := FD ( P ) ;

                   Distance := Distance + 1 ;

               FSI
           FTQ ;

       FSI ;

       Distance_feuille := Distance

   FIN
   ACTION Distance ( L )

   SOIT
       L , P DES LISTES DE ARB ;

       F UN ARB ;

   DEBUT
       P := L ;

       TQ SUIVANT ( P ) <> NIL
           ECRIRE ( Distance_feuille ( VALEUR ( P ) ) ) ;

           P := SUIVANT ( P )

       FTQ
   FIN
Utiliser un parcours récursif pour sauvegarder un arbre de recherche binaire dans un fichier TOF (Tableau ordonné fixe)
   ACTION Sauvegarde ( A )

   SOIT
       A UN ARB ;

   DEBUT
       SI A <> NIL
           APPEL Sauvegarde ( FG ( A ) ) ;
           AFF_STRUCT ( Buf , 1 , STRUCT ( Buf , 1 ) + 1 ) ;

           SI STRUCT ( Buf , 1 ) <= 3

               AFF_ELEMENT ( STRUCT ( Buf , 2 ) [ STRUCT ( Buf , 1 ) ] , INFO ( A ) ) ;

              {AFF_STRUCT ( Buf , 1 , Ind_buf ) ;}
           SINON
               AFF_STRUCT ( Buf , 1 , STRUCT ( Buf , 1 ) - 1 ) ;

               ECRIRESEQ ( F , Buf ) ;

               AFF_ENTETE ( F , 1 , ENTETE ( F , 1 ) + 1 ) ;

               AFF_STRUCT ( Buf , 1 , 1 ) ;

               AFF_ELEMENT ( STRUCT ( Buf , 2 ) [ 1 ] , INFO ( A ) ) ;

           FSI ;
           APPEL Sauvegarde ( FD ( A ) ) ;

       FSI
   FIN
   ACTION Afficher_fichier ;

   SOIT
       I UN ENTIER ;

       K UN ENTIER ;

   DEBUT
       OUVRIR ( F , 'test.txt' , 'A' ) ;

       K := 0 ;

       TQ NON FINFICH ( F )

           LIRESEQ ( F , Buf ) ;

           K := K + 1 ;

           ECRIRE ( 'Bloc ' , K , '=' ) ;

           POUR I := 1 , STRUCT ( Buf , 1 )

               ECRIRE ( ELEMENT ( STRUCT ( Buf , 2 ) [ I ] ) )

           FPOUR ;

       FTQ ;

       FERMER ( F )

   FIN
Comment restaurer l’arbre du fichier de telle sorte qu’il soit équilibré? 
Il s‘agit de lire le fichier et de reconstruire un arbre équilibré en mémoire.
On peut avoir les stratégies suivantes :

- utiliser la technique AVL ou arbres Rouge et Noir

- insérer dans l’ordre de parcours dichotomique des blocs du fichier, puisque ce dernier est trié.

